
Motivation
 The optimization modeling cycle

 Optimization modeling languages

 Introductory example

The AMPL company

AMPL’s users
 Commercial, government, research & teaching

 Two Edelman competition cases

Future directions
 More powerful interfaces

 More natural modeling
 Logical conditions
 Quadratic constraints

Steps
 Communicate with problem owner

 Build model

 Prepare data

 Generate optimization problem

 Submit problem to solver
 CPLEX, Gurobi, KNITRO, CONOPT, MINOS, . . .

 Report & analyze results

 Repeat!

Goals
 Do this quickly and reliably

 Get results before client loses interest

 Deploy for application

The Optimization Modeling Cycle

“We do not feel that the linear programming user’s 
most pressing need over the next few years is for a 
new optimizer that runs twice as fast on a machine 
that costs half as much (although this will probably 
happen).  Cost of optimization is just not the 
dominant barrier to LP model implementation.  

“The process required to manage the data, 
formulate and build the model, report on and 
analyze the results costs far more, and is much more 
of a barrier to effective use of LP, than the 
cost/performance of the optimizer.”

Krabek, Sjoquist, Sommer, 
“The APEX Systems: Past and Future.” 
SIGMAP Bulletin 29 (April 1980) 3-23.

What Makes This Hard?

Two forms of an optimization problem
 Modeler’s form

 Mathematical description, easy for people to work with

 Algorithm’s form
 Explicit data structure, easy for solvers to compute with

Idea of a modeling language
 A computer-readable modeler’s form

 You write optimization problems in a modeling language
 Computers translate to algorithm’s form for solution

Advantages of a modeling language
 Faster modeling cycles

 More reliable modeling and maintenance

Optimization Modeling Languages

Formulation concept
 Define data in terms of sets & parameters

 Analogous to database keys & records

 Define decision variables

 Minimize or maximize a function of decision variables

 Subject to equations or inequalities
that constrain the values of the variables

Advantages
 Familiar

 Powerful

 Implemented

Algebraic Modeling Languages

Features
 Algebraic modeling language

 Variety of data sources

 Connections to all solver features

 Interactive and scripted control

Advantages
 Powerful, general expressions

 Natural, easy-to-learn design

 Efficient processing scales well with problem size

8

The AMPL Modeling Language

Multicommodity transportation . . .
 Products available at factories

 Products needed at stores

 Plan shipments at lowest cost

. . . with practical restrictions
 Cost has fixed and variables parts

 Shipments cannot be too small

 Factories cannot serve too many stores

Given
ܱ Set of origins (factories)
ܦ Set of destinations (stores)
ܲ Set of products

and
ܽ௜௣ Amount available, for each ݅ ∈ ܱ and ݌ ∈ ܲ
௝ܾ௣ Amount required, for each ݆ ∈ ܦ and ݌ ∈ ܲ	
݈݆݅ Limit on total shipments, for each ݅ ∈ ܱ and ݆ ∈ ܦ

݌݆݅ܿ Shipping cost per unit, for each ݅ ∈ ܱ, ݆ ∈ ,ܦ ݌ ∈ ܲ
݆݀݅ Fixed cost for shipping any amount from ݅ ∈ ܱ	to ݆ ∈ ܦ

ݏ Minimum total size of any shipment
݊ Maximum number of destinations served by any origin

Multicommodity Transportation

Determine
݌݆݅ܺ Amount of each ݌ ∈ ܲ to be shipped from ݅ ∈ ܱ	to ݆ ∈ ܦ
ܻ݆݅ 1 if any product is shipped from ݅ ∈ ܱ	to ݆ ∈ ܦ

0 otherwise

to minimize
∑ ∑ ∑ ܿ௜௝௣௣∈௉௝∈஽ ௜ܺ௝௣௜∈ை ൅	∑ ∑ ݀௜௝௝∈஽ ௜ܻ௝௜∈ை

Total variable cost plus total fixed cost

12

Mathematical Formulation
Multicommodity Transportation

Subject to
∑ ௜ܺ௝௣௝∈஽ ൑ ܽ௜௣ for all ݅ ∈ ܱ, ݌ ∈ ܲ

Total shipments of product ݌ out of origin ݅
must not exceed availability

∑ ௜ܺ௝௣௜∈ை ൌ ௝ܾ௣ for all ݆ ∈ ݌ ,ܦ ∈ ܲ

Total shipments of product	݌ into destination ݆
must satisfy requirements

Mathematical Formulation
Multicommodity Transportation

Subject to
∑ ௜ܺ௝௣௣∈௉ ൑ ݈௜௝ ௜ܻ௝ for all ݅ ∈ ܱ, ݆ ∈ ܦ

When there are shipments from origin ݅ to destination ݆,
the total may not exceed the limit, and ௜ܻ௝ must be 1

∑ ௜ܺ௝௣௣∈௉ ൒ ݏ ௜ܻ௝ for all ݅ ∈ ܱ, ݆ ∈ ܦ

When there are shipments from origin ݅ to destination ݆, 
the total amount of shipments must be at least ݏ

∑ ௜ܻ௝௝∈஽ ൑ ݊ for all ݅ ∈ ܱ

Number of destinations served by origin ݅
must be as most ݊

Mathematical Formulation
Multicommodity Transportation

AMPL Formulation

Symbolic data

set ORIG;   # origins
set DEST;   # destinations
set PROD;   # products

param supply {ORIG,PROD} >= 0;  # availabilities at origins
param demand {DEST,PROD} >= 0;  # requirements at destinations
param limit {ORIG,DEST} >= 0;   # capacities of links

param vcost {ORIG,DEST,PROD} >= 0; # variable shipment cost
param fcost {ORIG,DEST} > 0;       # fixed usage cost

param minload >= 0;             # minimum shipment size
param maxserve integer > 0;     # maximum destinations served

Multicommodity Transportation

AMPL Formulation

Symbolic model: variables and objective

var Trans {ORIG,DEST,PROD} >= 0;   # actual units to be shipped

var Use {ORIG, DEST} binary;       # 1 if link used, 0 otherwise

minimize Total_Cost:

sum {i in ORIG, j in DEST, p in PROD} vcost[i,j,p] * Trans[i,j,p]

+ sum {i in ORIG, j in DEST} fcost[i,j] * Use[i,j];

Multicommodity Transportation

∑ ∑ ∑ ܿ௜௝௣௣∈௉௝∈஽ ௜ܺ௝௣௜∈ை ൅	∑ ∑ ݀௜௝௝∈஽ ௜ܻ௝௜∈ை

AMPL Formulation

Symbolic model: constraint

subject to Supply {i in ORIG, p in PROD}:

sum {j in DEST} Trans[i,j,p] <= supply[i,p];

Multicommodity Transportation

∑ ௜ܺ௝௣௝∈஽ ൑ ܽ௜௣,		for all ݅ ∈ ݌ ,ܱ ∈ ܲ



R. Fourer and D.M. Gay, New & Forthcoming Developments in AMPL
INFORMS Conf on Analytics — 15-17 April 2012 — Track 22, Software Tutorials 18

AMPL Formulation

Symbolic model: constraints

subject to Supply {i in ORIG, p in PROD}:

sum {j in DEST} Trans[i,j,p] <= supply[i,p];

subject to Demand {j in DEST, p in PROD}:

sum {i in ORIG} Trans[i,j,p] = demand[j,p];

subject to Multi {i in ORIG, j in DEST}:

sum {p in PROD} Trans[i,j,p] <= limit[i,j] * Use[i,j];

subject to Min_Ship {i in ORIG, j in DEST}:

sum {p in PROD} Trans[i,j,p] >= minload * Use[i,j];

subject to Max_Serve {i in ORIG}:

sum {j in DEST} Use[i,j] <= maxserve;

Multicommodity Transportation

AMPL Formulation

Explicit data independent of symbolic model

set ORIG := GARY CLEV PITT ;
set DEST := FRA DET LAN WIN STL FRE LAF ;
set PROD := bands coils plate ;

param supply (tr):  GARY   CLEV   PITT :=
bands    400    700    800
coils    800   1600   1800
plate    200    300    300 ;

param demand (tr):
FRA   DET   LAN   WIN   STL   FRE   LAF :=

bands   300   300   100    75   650   225   250
coils   500   750   400   250   950   850   500
plate   100   100     0    50   200   100   250 ;

param limit default 625 ;

param minload := 375 ;
param maxserve := 5 ;

Multicommodity Transportation

AMPL Formulation

Explicit data (continued)

param vcost :=

[*,*,bands]:  FRA  DET  LAN  WIN  STL  FRE  LAF :=
GARY    30   10    8   10   11   71    6
CLEV    22    7   10    7   21   82   13
PITT    19   11   12   10   25   83   15

[*,*,coils]:  FRA  DET  LAN  WIN  STL  FRE  LAF :=
GARY    39   14   11   14   16   82    8
CLEV    27    9   12    9   26   95   17
PITT    24   14   17   13   28   99   20

[*,*,plate]:  FRA  DET  LAN  WIN  STL  FRE  LAF :=
GARY    41   15   12   16   17   86    8
CLEV    29    9   13    9   28   99   18
PITT    26   14   17   13   31  104   20 ;

param fcost:   FRA  DET  LAN  WIN  STL  FRE  LAF :=
GARY  3000 1200 1200 1200 2500 3500 2500
CLEV  2000 1000 1500 1200 2500 3000 2200
PITT  2000 1200 1500 1500 2500 3500 2200 ;

Multicommodity Transportation

AMPL Solution

Model + data = problem instance to be solved

ampl: model multmipG.mod;
ampl: data multmipG.dat;

ampl: option solver gurobi;

ampl: solve;

Gurobi 4.6.0: optimal solution; objective 235625
404 simplex iterations
45 branch-and-cut nodes

ampl: display Use;

Use [*,*]

:    DET FRA FRE LAF LAN STL WIN    :=
CLEV   1   1   1   0   1   1   0
GARY   0   0   0   1   0   1   1
PITT   1   1   1   1   0   1   0
;

Multicommodity Transportation

AMPL Solution

Solver choice independent of model and data

ampl: model multmipG.mod;
ampl: data multmipG.dat;

ampl: option solver cplex;

ampl: solve;

CPLEX 12.4.0.0: optimal integer solution; objective 235625
394 MIP simplex iterations
41 branch-and-bound nodes

ampl: display Use;

Use [*,*]

:    DET FRA FRE LAF LAN STL WIN    :=
CLEV   1   1   1   0   1   1   0
GARY   0   0   0   1   0   1   1
PITT   1   1   1   1   0   1   0
;

Multicommodity Transportation

AMPL Solution

Examine results

ampl: display {i in ORIG, j in DEST}
ampl?    sum {p in PROD} Trans[i,j,p] / limit[i,j];

:      DET    FRA    FRE    LAF    LAN   STL    WIN    :=
CLEV   1      0.6    0.88   0      0.8   0.88   0
GARY   0      0      0      0.64   0     1      0.6
PITT   0.84   0.84   1      0.96   0     1      0
;

ampl: display Max_Serve.body;

CLEV  5
GARY  3
PITT  5
;

ampl: display TotalCost,
ampl?    sum {i in ORIG, j in DEST} fcost[i,j] * Use[i,j];

TotalCost = 235625
sum {i in ORIG, j in DEST} fcost[i,j]*Use[i,j] = 27600

Multicommodity Transportation

Indexed over sets of pairs and triples

AMPL “Sparse” Network

set ORIG;   # origins
set DEST;   # destinations
set PROD;   # products

set SHIP within {ORIG,DEST,PROD};

# (i,j,p) in SHIP ==> can ship p from i to j

set LINK = setof {(i,j,p) in SHIP} (i,j);

# (i,j) in LINK ==> can ship some products from i to j

...........

var Trans {SHIP} >= 0;   # actual units to be shipped

var Use {LINK} binary;   # 1 if link used, 0 otherwise

minimize Total_Cost:

sum {(i,j,p) in SHIP} vcost[i,j,p] * Trans[i,j,p]

+ sum {(i,j) in LINK} fcost[i,j] * Use[i,j];

Multicommodity Transportation

AMPL Scripting

Script to test sensitivity to serve limit

model multmipG.mod;
data multmipG.dat;

option solver gurobi;

for {m in 7..1 by -1} {
let maxserve := m;
solve;
if solve_result = 'infeasible' then break;
display maxserve, Max_Serve.body;

}

Multicommodity Transportation

AMPL Scripting

Run showing sensitivity to serve limit

ampl: include multmipServ.run;

Gurobi 4.6.0: optimal solution; objective 233150

maxserve = 7

CLEV 5   GARY 3   PITT 6

Gurobi 4.6.0: optimal solution; objective 233150

maxserve = 6

CLEV 5   GARY 3   PITT 6

Gurobi 4.6.0: optimal solution; objective 235625

maxserve = 5

CLEV 5   GARY 3   PITT 5

Gurobi 4.6.0: infeasible

Multicommodity Transportation

AMPL Scripting

Script to generate n best solutions

param nSols default 0;
param maxSols;

model multmipG.mod;
data multmipG.dat;

set USED {1..nSols} within {ORIG,DEST};

subject to exclude {k in 1..nSols}:

sum {(i,j) in USED[k]} (1-Use[i,j]) + 
sum {(i,j) in {ORIG,DEST} diff USED[k]} Use[i,j] >= 1;

option solver gurobi;

repeat {
solve;
display Use;
let nSols := nSols + 1;
let USED[nSols] := {i in ORIG, j in DEST: Use[i,j] > .5};

} until nSols = maxSols;

Multicommodity Transportation

AMPL Scripting

Run showing 3 best solutions

ampl: include multmipBest.run;

Gurobi 4.6.0: optimal solution; objective 235625

:    DET FRA FRE LAF LAN STL WIN    :=
CLEV   1   1   1   0   1   1   0
GARY   0   0   0   1   0   1   1
PITT   1   1   1   1   0   1   0 ;

Gurobi 4.6.0: optimal solution; objective 237125

:    DET FRA FRE LAF LAN STL WIN    :=
CLEV   1   1   1   1   0   1   0
GARY   0   0   0   1   0   1   1
PITT   1   1   1   0   1   1   0 ;

Gurobi 4.6.0: optimal solution; objective 238225

:    DET FRA FRE LAF LAN STL WIN    :=

CLEV   1   0   1   0   1   1   1
GARY   0   1   0   1   0   1   0
PITT   1   1   1   1   0   1   0 ;

Multicommodity Transportation

AMPL history

The AMPL team

Recent developments
 Business

 Academic

34

The AMPL Company

Origins
 AMPL developed at AT&T Bell Laboratories (1986)

 Robert Fourer, David M. Gay, Brian W. Kernighan

 AMPL sold through distributors (1993)

 AMPL Optimization company formed (2002)

Writings
 “A Modeling Language for Mathematical Programming.”  

Management Science 36 (1990) 519–554.

 The AMPL book

35

AMPL History
AMPL Company

1993 2003

Robert Fourer
 Managing partner

David M. Gay
 Managing partner

William M. Wells
 Director of business development (joined 2010)

Victor Zverovich
 Optimization software development and support

36

The AMPL Team
AMPL Company

AMPL intellectual property
 Full rights acquired from Alcatel-Lucent USA

 corporate parent of Bell Laboratories

 More flexible licensing terms available

CPLEX & Gurobi for AMPL
 CPLEX sales transferred from IBM to AMPL Optimization

 Full lineup of licensing arrangements available

AMPL distributors
 New for Japan: October Sky Co., Ltd. 
 Others continue active

 Gurobi, Ziena/Artelys
 MOSEK, TOMLAB
 OptiRisk

37

Business Developments

Highly discounted prices for academic use
 AMPL

 Nonlinear solvers:  KNITRO, MINOS, SNOPT, CONOPT

Free MIP solvers to academic users
 Gurobi & CPLEX

 1-year licenses

Free AMPL & solvers for courses
 One-page application (www.ampl.com/courses.html)

 Single file for distribution to students

 Streamlined installation — no license file

 Expires when the course is over

38

Academic Developments

Business
 Customer relationships

 Customer areas

 Project examples

Government

Academic

39

AMPL’s Users

Internal projects
 We supply software & answer a few questions

 Company’s employees build the models

Training and consulting
 Available on request

40

Business Customer Relationships
AMPL’s Users

Transportation
 Air, rail, truck

Production
 Planning

 steel
 automotive

 Supply chain
 consumer products

Finance
 Investment banking

 Insurance

41

Business Customer Areas
AMPL’s Users

Natural resources
 Electric power

 Gas distribution

 Mining

Information technology
 Telecommunications

 Internet services

Consulting practices
 Management

 Industrial engineering

Two award-winning projects
 ZARA (clothing retailing)

 Norske Skog (paper manufacturer)
. . . finalists for Edelman Award

for practice of Operations Research

42

Business Customer Examples
AMPL’s Users

Optimization of worldwide shipments
 Legacy process

 Piecewise-linear AMPL model with integer variables

 Run once for each product each week

 Decides how much of each size to ship to each store

 Increases sales 3-4%

43

ZARA
AMPL’s Users

New Process

Given
ܵ ൌ Sା ∪ Sି

Set of sizes partitioned into major & regular sizes
ܬ Set of stores

and
௦ܹ Inventory of size ݏ available at the warehouse

௦௝ܫ Inventory of size ݏ available	in	store	݆
௝ܲ Selling price in store ݆

ܭ Aggressiveness factor (value of inventory remaining
in the warehouse after the current shipments)

௦௝ߣ Demand rate for size ݏ in store ݆
௦ܰ௝ Approximation set for size ݏ in the inventory-to-sales

function approximation for store ݆

44

ZARA’s Formulation
AMPL’s Users



R. Fourer and D.M. Gay, New & Forthcoming Developments in AMPL
INFORMS Conf on Analytics — 15-17 April 2012 — Track 22, Software Tutorials

Determine
௦௝ݔ (integer) shipment quantity of each size ݏ ∈ ܵ

to each store ݆ ∈ ܬ for the current replenishment period

௝ݖ (൒ 0) approximate expected sales across all sizes 
in each store ݆ ∈ ܬ for the current period

to maximize
∑ ௝ܲݖ௝௝∈௃ ൅ ∑ܭ ሺ ௦ܹ െ ∑ ௦௝ሻ௝∈௃௦∈ௌݔ

Total sales plus value of items remaining in warehouse

subject to
∑ ௦௝ݔ ൑ ௦ܹ௝∈௃ for all ݏ ∈ ܵ

Total shipments of size ݏ
must not exceed amount available in warehouse

45

ZARA’s Formulation
AMPL’s Users

and subject to
௝ݖ ൑ ሺ∑ ௝ݕ௦௝ሻߣ ൅௦∈ௌశ ∑ ௦௝௦∈ௌషݒ௦௝ߣ for all ݆ ∈ ܬ

௝ݕ ൑ ܽ௜ߣ௦௝ ௦௝ܫ ൅ ௦௝ݔ െ ݅ ൅ ܾ௜ߣ௦௝ for all ݆ ∈ ݏ ,ܬ ∈ ܵା, ݅ ∈ ௦ܰ௝

௦௝ݒ ൑ ܽ௜ߣ௦௝ ௦௝ܫ ൅ ௦௝ݔ െ ݅ ൅ ܾ௜ߣ௦௝ for all ݆ ∈ ݏ ,ܬ ∈ ܵି, ݅ ∈ ௦ܰ௝

௦௝ݒ ൑ ௝ݕ for all ݆ ∈ ݏ ,ܬ ∈ ܵି

Relationship between sales
and store inventory after shipments

46

ZARA’s Formulation
AMPL’s Users

Optimization of production and distribution
 Australasia

 Europe
 640 binary variables
 524,000 continuous variables
 33,000 constraints

Optimization of shutdown decisions worldwide
 Multiple scenarios

 Numerous sensitivity analyses

 Key role of AMPL models
 Implemented in a few weeks
 Modified to analyze alternatives
 Run interactively at meetings

Given sets
ܰ Number of mills
௡ܯ Set of machines at mill ܰ
ܯ Number of paper machines
ܬ Number of products
ܮ Number of raw material sources
ܴ Number of raw materials
ܭ Number of customers
ܲ Number of recipes

Given capital parameters
݈௡ fixed cost of mill ݊ running for one period

(excluding machine fixed costs)

௠݂ fixed cost of machine ݉ running for one period

௠ߠ proportion of fixed running costs saved from
a temporary shutdown on machine ݉

௠ݍ minimum time that machine ݉ must be shut
before savings accrue

߶௠ amortized cost of a permanent closure of machine ݉

and operating parameters
݃௠௝௞ variable freight cost for shipping product ݆	

from machine ݉ to customer ݇
ܽ௠௝௣ capacity of machine ݉

making product ݆ using recipe ݌
ܿ௠௝௣ variable cost incurred by producing 

one tonne of product ݆ using recipe ݌ on machine ݉
݄௠௝௥௣ tonnes of raw material ݎ required to make

one tonne of product ݆ using recipe ݌ on machine ݉
௠௥௟ߨ procurement, transportation, and process cost

of raw material ݎ from source ݈ for machine ݉

௥ܹ௟ supply of raw material ݎ at source ݈

௝݀௞ demand for product ݆ by customer ݇
௝௞ݏ sales price for product ݆ by customer ݇

Make capital decisions
௡ߜ 1 if mill ݊ closes, 0 otherwise

௠ߤ 1 if machine ݉ shuts down permanently, 0 otherwise

௠ݑ time that machine ݉ has been shut down

௠ߦ 1 if machine ݉ has been shut down long enough
to accrue savings, 0 otherwise

௠ݒ time that qualifies for savings on machine ݉

and operating decisions
௠௝௣ݔ tonnes of product ݆ made on machine ݉

using recipe ݌
௠௝௞ݕ tonnes of product ݆ made on machine ݉ and

delivered to customer ݇
௠௥௟ݓ tonnes of raw material ݎ from source ݈

used by machine ݉
௠௣ߪ 1 if recipe ݌ is used on machine ݉, 0 otherwise

Maximize

∑ ∑ ∑ ሺݏ௝௞ െ ݃௠௝௞௄
௞ୀଵ ሻݕ௠௝௞ െ ∑ ܿ௠௝௣ݔ௠௝௣௉

௣ୀଵ
௃
௝ୀଵ

ெ
௠ୀଵ

െ∑ ∑ ∑ ௠௥௟ݓ௠௥௟ߨ
ோ
௥ୀଵ

௅
௟ୀଵ

ெ
௠ୀଵ

൅∑ ௠ߠ ௠݂ݒ௠ெ
௠ୀଵ

െ∑ ݈௡ 1 െ ௡ߜ ൅ ௡ேߜ௡ߣ
௡ୀଵ

െ∑ ௠݂ 1 െ ௠ߤ ൅ ߶௠ߤ௠ெ
௠ୀଵ

Income from sales,
minus raw material, production and distribution costs,
plus savings from shutdowns,
minus fixed operating and shutdown costs

Subject to

∑ ∑ ௫೘ೕ೛

௔೘ೕ೛
ൌ 1 െ ௠௉ݑ

௣ୀଵ
௃
௝ୀଵ for ݉ ൌ 1, . . . , ܯ

Capacity used equals capacity available

∑ ௠௝௞ݕ ൌ ∑ ௠௝௣௉ݔ
௣ୀଵ

௄
௞ୀଵ for	݆ ൌ 1, . . . , ,ܬ ݉ ൌ 1, . . . , ܯ

Amounts produced equal amounts shipped

∑ ௠௝௞ݕ ൑	ெ
௠ୀଵ ௝݀௞ for	݆ ൌ 1, . . . , ,ܬ ݇ ൌ 1, . . . , ܭ

Amounts produced do not exceed demand

∑ ∑ ݄௠௝௥௣ݔ௠௝௣ ൌ ∑ ௠௥௟ݓ
௅
௟ୀଵ

௉
௣ୀଵ

௃
௝ୀଵ for	݉ ൌ 1, . . . , ,ܯ ݎ ൌ 1, . . . , ܴ

Raw material used equals raw material purchased

∑ ௠௥௟ݓ ൑	ெ
௠ୀଵ ௥ܹ௟ for	݈ ൌ 1, . . . , ,ܮ ݎ ൌ 1, . . . , ܴ

Raw material purchased does not exceed amount available

and subject to
∑ ௠௣ߪ ൌ 1 െ ௠௉ߤ
௣ୀଵ for	݉ ൌ 1, . . . , ܯ

௠௝௣ݔ ൑ ܽ௠௝௣ߪ௠௝௣ for	݆ ൌ 1, . . . , ,ܬ ݉ ൌ 1, . . . , ,ܯ ݌ ൌ 1, . . . , ܲ

௡ߜ ൑ ௠ߤ for	݉ ∈ ,௡ܯ ݊ ൌ 1, . . . , ܰ
௠ݒ ൑ ௠ߦ	 for	݉ ൌ 1, . . . , ܯ
௠ݒ ൑ 	1 െ ௠ߤ for	݉ ൌ 1, . . . , ܯ
௠ݒ ൑ ௠ݑ	 െ ௠ߦ௠ݍ for	݉ ൌ 1, . . . , ܯ

Definitions of zero-one variables

Financial agencies
 United States

 Canada

 Sweden

U.S. departments
 Census Bureau

 Army Corps of Engineers

U.S. research centers
 Argonne National Laboratory

 Sandia National Laboratories

 Lawrence Berkeley Laboratory

Research
 Over 250 university installations worldwide

 Nearly 1000 citations in scientific papers
 engineering, science, economics, management

Teaching
 Linear & nonlinear optimization

 Graph optimization
 Stochastic programming

 Operations Research

 Specialized courses
 Supply chain modeling
 Electric power system planning
 Transportation logistics
 Communication network design & algorithms

More powerful interfaces
 Enhanced user interfaces (IDEs)

 Callable interfaces for deployment (APIs)

More natural modeling
 Logical conditions

 Quadratic constraints

Logical Conditions

Minimum-shipment constraints
 From each origin to each destination, either

ship nothing or ship at least minload units

Conventional linear mixed-integer formulation

var Trans {ORIG,DEST,PROD} >= 0;

var Use {ORIG, DEST} binary;

....... 

subject to Multi {i in ORIG, j in DEST}:

sum {p in PROD} Trans[i,j,p] <= limit[i,j] * Use[i,j];

subject to Min_Ship {i in ORIG, j in DEST}:

sum {p in PROD} Trans[i,j,p] >= minload * Use[i,j];

More Natural Modeling

Zero-One Alternatives

Mixed-integer formulation using implications

subject to Multi {i in ORIG, j in DEST}:

Use[i,j] = 0 ==> sum {p in PROD} Trans[i,j,p] = 0;

subject to Min_Ship {i in ORIG, j in DEST}:

Use[i,j] = 1  ==>

minload <= sum {p in PROD} Trans[i,j,p] <= limit[i,j];

Logical Conditions

. . . implemented in CPLEX

subject to Multi_Min_Ship {i in ORIG, j in DEST}:

Use[i,j] = 1  ==>

minload <= sum {p in PROD} Trans[i,j,p] <= limit[i,j]

else sum {p in PROD} Trans[i,j,p] = 0;

Non-Zero-One Alternatives

Disjunctive constraint

subject to Multi {i in ORIG, j in DEST}:

sum {p in PROD} Trans[i,j,p] = 0  or

minload <= sum {p in PROD} Trans[i,j,p] <= limit[i,j];

Logical Conditions

Discontinuous domain

var SumTrans {i in ORIG, j in DEST} 

in {0} union interval[minload,limit[i,j]];

.......

subject to Multi {i in ORIG, j in DEST}:

SumTrans[i,j] = sum {p in PROD} Trans[i,j,p];

Given a traffic network
ܰ Set of nodes representing intersections

݁ Entrance to network

݂ Exit from network

	ܣ ⊆ 	ܰ ∪ ሼ݁ሽ 	ൈ 	ܰ ∪ ሼ݂ሽ
Set of arcs representing road links

with associated data
ܾ݆݅ Base travel time for each road link ሺ݅, ݆ሻ ∈ ܣ
݆݅ݏ Traffic sensitivity for each road link ሺ݅, ݆ሻ ∈ ܣ
݆ܿ݅ Capacity for each road link ሺ݅, ݆ሻ ∈ ܣ
ܶ Desired throughput from ݁ to ݂

62

Quadratic Constraints
More Natural Modeling



R. Fourer and D.M. Gay, New & Forthcoming Developments in AMPL
INFORMS Conf on Analytics — 15-17 April 2012 — Track 22, Software Tutorials

Determine
݆݅ݔ Traffic flow through road link ሺ݅, ݆ሻ ∈ ܣ
݆݅ݐ Actual travel time on road link ሺ݅, ݆ሻ ∈ ܣ

to minimize

Average travel time from e to f

Subject to

Formulation (cont’d)

Travel times increase as flow approaches capacity

Flow out equals flow in at any intersection

Flow into the entrance equals the specified throughput

Traffic Network

AMPL Formulation

Symbolic data

set INTERS;          # intersections (network nodes)

param EN symbolic;   # entrance
param EX symbolic;   # exit

check {EN,EX} not within INTERS;

set ROADS within {INTERS union {EN}} cross {INTERS union {EX}};

# road links (network arcs)

param base {ROADS} > 0;  # base travel times
param sens {ROADS} > 0;  # traffic sensitivities
param cap {ROADS} > 0;   # capacities

param through > 0;       # throughput

Traffic Network



R. Fourer and D.M. Gay, New & Forthcoming Developments in AMPL
INFORMS Conf on Analytics — 15-17 April 2012 — Track 22, Software Tutorials 66

AMPL Formulation (cont’d)

Symbolic model

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Time {ROADS} >= 0;

minimize Avg_Time:
(sum {(i,j) in ROADS} Time[i,j] * Flow[i,j]) / through;

subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = base[i,j] + (sens[i,j]*Flow[i,j]) / (1-Flow[i,j]/cap[i,j]);

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Data

Explicit data independent of symbolic model

set INTERS := b c ;

param EN := a ;
param EX := d ;

param: ROADS: base cap sens :=
a b    4   10   .1
a c    1   12   .7
c b    2   20   .9
b d    1   15   .5
c d    6   10   .1 ;

param through := 20 ;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX?

ampl: model traffic.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.3.0.0: 
Constraint _scon[1] is not convex quadratic 
since it is an equality constraint.

Traffic Network

AMPL Solution (cont’d)

Look at the model again . . .

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Time {ROADS} >= 0;

minimize Avg_Time:
(sum {(i,j) in ROADS} Time[i,j] * Flow[i,j]) / through;

subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = base[i,j] + (sens[i,j]*Flow[i,j]) / (1-Flow[i,j]/cap[i,j]);

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Quadratically constrained reformulation

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Delay {ROADS} >= 0;

minimize Avg_Time:
sum {(i,j) in ROADS} (base[i,j]*Flow[i,j] + Delay[i,j]) / through;

subject to Delay_Def {(i,j) in ROADS}:
sens[i,j] * Flow[i,j]^2 <= (1 - Flow[i,j]/cap[i,j]) * Delay[i,j];

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX?

ampl: model trafficQUAD.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.3.0.0: 
QP Hessian is not positive semi-definite.

Traffic Network

AMPL Solution (cont’d)

Simple conic quadratic reformulation

var Flow {(i,j) in ROADS} >= 0, <= .9999 * cap[i,j];
var Delay {ROADS} >= 0;
var Slack {ROADS} >= 0;

minimize Avg_Time:
sum {(i,j) in ROADS} (base[i,j]*Flow[i,j] + Delay[i,j]) / through;

subject to Delay_Def {(i,j) in ROADS}:
sens[i,j] * Flow[i,j]^2 <= Slack[i,j] * Delay[i,j];

subject to Slack_Def {(i,j) in ROADS}:
Slack[i,j] = 1 - Flow[i,j]/cap[i,j];

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,i];

subject to Balance_Enter:
sum{(EN,j) in ROADS} Flow[EN,j] = through;

Traffic Network

AMPL Solution (cont’d)

Model + data = problem to solve, using CPLEX!

ampl: model trafficSOC.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.3.0.0: primal optimal; objective 61.04693968
15 barrier iterations

ampl: display Flow;

Flow :=
a b    9.55175
a c   10.4482
b d   11.0044
c b    1.45264
c d    8.99561
;

Traffic Network

AMPL Solution (cont’d)

Same with integer-valued variables

ampl: solve;

CPLEX 12.3.0.0: optimal integer solution within mipgap or absmipgap; 
objective 76.26375017

19 MIP barrier iterations
0 branch-and-bound nodes

ampl: display Flow;

Flow :=
a b    9
a c   11
b d   11
c b    2
c d    9
;

var Flow {(i,j) in ROADS} integer >= 0, <= .9999 * cap[i,j];

Traffic Network

Problem types
 Elliptical: quadratic programs (QPs)

 Conic: second-order cone programs (SOCPs)

Current situation
 Each solver recognizes some elementary forms

 Modeler must convert to these forms

Goal
 Recognize many equivalent forms

 Automatically convert to a canonical form

 Further convert as necessary for each solver

